DECUS
PROGRAM LIBRARY

DECUS NO.	8-772
TITLE	OS/8 COMPATIBLE VC8-E HANDLER FOR MASS STORAGE SYSTEMS
AUTHOR	Stephan V. Bechtolsheime, Ulrich Gschrei
COMPANY	Submitted by: L. von Lindern Max-Planck Institut fuer Psychiatrie Munich, West Germany
DATE	April 5, 1975
SOURCE LANGUAGE	PAL-8

GENERAL INFORMATION

Object Computer(s) PDP-8/E Source Computer (if different) _____

File Name PARAM.PA, HELP.PA Version No. _____

Title OS/8 Compatible VC8E Handler for Mass Storage Systems

Author Stephan von Bechtolsheime, Ulrich Gschrei

Submitter (if other than author) L. von Lindern

Affiliation Max-Planck Institut fuer Psychiatrie

Address 8 Munchen 40, Kraepelinstr. 2

Country W. Germany

Monitor/Operating System OS/8 V3 DEC No. _____

Core Storage Required ≥ 12K Starting Address _____

Peripherals Required VC8E interface and storage scope 611 or 613 Tektrx.

Other Software Required _____ DEC or DECUS No. _____

Source Language PAL-8 Category Utility; Display

Restrictions, Deficiencies, Problems Support program HELP must be loaded into 4 pages which are not accepted by a running program

Date of Planned or Possible Future Revisions _____

TAPES AVAILABLE

Paper Tapes Object Binary ☐ Object ASCII ☐ Source ☐ Other _____

DECtape ☒ LINCtape ☐ Format _____ Magtape: 7 Track ☐ 9 Track ☐ BPI _____

Object Files ☐ Source Files ☐ Documentation Files ☐ Other _____

ABSTRACT

See page 1 of write-up

OS/8 Compatible VC8/E Handler
For Mass Storage Systems

DECUS Program Library Write-up DECUS NO. 8-772

OS/8 Compatible VC8-E Handler for

Mass Storage Systems

Stephan v. Bechtolsheim
Ulrich Gschrei

Max-Planck-Institut fuer Psychiatrie
Kraepelinstr. 2
8 München 40
Western Germany

April 1975

1. Introduction

An OS-8 (12K) compatible scope handler is described for a PDP-8/E with a VC8-E interface for the storage scope Tektronix 611 or 613 and EAE option. It can be assigned like a lineprinter. No additional hardware is necessary. Characters display is via 5x7 matrix and effected by a one page handler cooperating with a co-resident program of four pages containing character table and display organisation (PAL8). Arbitrary character position in Fortran 2 is possible. Furthermore for the systemprogram Edit an option is implemented to direct TTY responses to the scope.

DECTAPE

```
DA
SATURDAY APRIL  5, 1975

.DIR DTA0:

 05-APR-75

PARAM .PA    6 05-APR-75
HELP  .PA   38 05-APR-75
HAVC8E.PA   15 05-APR-75
LOXY  .FT    4 05-APR-75

 667 FREE BLOCKS
```

I N D E X

2. Description of the Programs

As a handler must not exceed two pages, but the character table itself occupies one and a half page, we divided up the program into two parts.
Part one is the handler itself with two entries OUTP and DISP.
Part two is program HELP, which generates characters by a charactertable and controls length and number of lines. It must not be overwritten by a running program.

Parameter File PARAM

The handler and program HELP are translated together with a parameter file, which for instance must be edited by the user to specify starting address and field of program HELP. For details refer to the parameter file and to the following text.

2.1 The VC8-E Handler

device name	OUTP	DISP
group name	VC8E	
device number	50_8	51_8
relative entry	0	113_8
length	one page	

Function of the Handler

The VC8-E handler
(a) gets the arguments from the calling program, i.e. buffer address and buffer field, return address, word count;
(b) unpacks characters from the input buffer and controls the wordcount;
(c) calls program HELP to position and to display the characters.

Entry OUTP

This entry is used normally. On the first call of a program to the handler the screen is erased.

Entry DISP

The display of a program starts, where the previous one has stopped, if this entry is used. For further use refer to 4.1.2 (writing on arbitrary positions).

2.2 Program HELP

Function of Program HELP

(a) display the characters fetched from the handler by means of a character table;
(b) controls length and number of lines;
(c) reacts on control characters, e.g. formfeed and linefeed.

2.2.1 Entries tp Program HELP

Program HELP is loaded into the highest available field. Modify HEL=30 in the parameter file respectively. The starting address is selected by the symbol STRTHL (also in the parameter file).

Three entries are used:

name (=label in HELP)	absolute address	function
LOAD	STRTHL	used by the system to load program HELP into core
HAND	STRTHL+4	entry point for the handler
CHONE	STRTHL+7oo	used to display a single character; the handler is not involved

Entry LOAD

is specified as starting address to the system. The command R HELP acts as a load command: as soon as program HELP has been loaded and started, it transfers control back to the monitor.

Entry HAND

the handler executes a JMS to this symbolic address to call Program HELP, which fetches three unpacked characters. The screen is erased, if the MQ-register has been set to zero (entry OUTP of the handler uses this facility). After display program HELP returns to symbolic address RETB in the handler. RETA is used, if CTRL/Z has been found in the buffer(buffer end) or has been read from the keyboard executing subroutine FF of program HELP.

Entry CHONE

This entry is used to display a single character stored in the AC. Instead of printing on the teletype output can be listed on scope (in every PAL8 program for instance). For more details refer to the parameter file. CTRL/Z must never be typed.

2.2.2 Functional Elements of Program HELP

(1) Subroutine LFCR

executes linefeed and carriage return; if screen is full, subroutine FF is called.

(2) subroutine ERASE

erases screen and sets beam on top of page. If the hardware erase function of the VC8-E point plot display control is installed, edit ERFUNC=1 in the parameter file, else ERFUNC=0.

(3) Subroutine FF

executes formfeed in two diffrent ways, which can be selected in the parameter file:

AUTOFF=0

the subroutine reads from the keyboard and erases screen (using subroutine ERASE), if space has been typed. CTRL/Z forces an immediate return to the handler (symbolic address RETA); thus the rest of the buffer is ignored. CTRL/C calls the monitor. Other characters are ignored.

AUTOFF=1 (automatic page processing)

Formfeed is executed automatically after a delay of 2 seconds. It can be compared with turning over the leaves of a book. Typing CTRL/S prolongs the delay until space is typed. The meaning of CTRL/C and CTRL/Z is as given above.

(4) Subroutine SPADIS

is used to 'display' a space character: only the x-position of the beam is incremented.

(5) Subroutine TOOLON

calls subroutine LFCR, if current line is too long.

(6) Subroutine CHAR

is the essential character generator derived from the character display routine supplied by J.A. Flavell, Max-Planck-Institut fuer Physik und Astrophysik, Muenchen.

(a) serve control characters

character	8 bit ASCII	Reaction of subroutine CHAR
space	240	JMS SPADIS
FF	214	(1)
CTRL/Z	232	buffer end found: program HELP returns to symbolic address RETA in the handler
LF	212	ignored, as always combined with CR
CR	215	JMS LFCR
	012	is ignored (occurs when listing directories by PIP)
TAB	211	an appropriate number of spaces are 'displayed'
rubout	377	is ignored

(1) Reaction of subroutine CHAR, when FF read in buffer

Conditional assembly defined through the variable BUFFFF in the parameter file determines the reaction of subroutine CHAR:

BUFFFF=1

A formfeed character causes a JMS FF, i.e. formfeed is executed.

BUFFFF=0

The formfeed character is treated identically to the linefeed character, i.e. subroutine LFCR is called. Thus subroutine FF is called only by subroutine LFCR, if screen is full.

(b) the characters are displayed, if none of these listed above, by means of a 5x7 dot matrix.

3. How to Use It

3.1 System Implementation

Program HELP is normally loaded into the highest available field occupiing 4 pages. But it is very difficult to give optimal values for its starting address. If BATCH or FRTS are not used, specify STRTHL=7000 in the parameter file (program HELP then occupies locations 7000-7777). If BATCH is run, specify STRTHL=4000 or less (but always a page boundary); so the BATCH monitor will not be destroyed. In any case a running program must not load into locations of program HELP or use them.

Inserting the VC8-E Handler into the System

First translate it:

```
.R PAL8
*HAVC8E<PARAM,HAVC8E
```

Then load it into the system and make it an active handler by program BUILD:

```
.R BUILD
$ LOAD HAVC8E
$ IN VC8E,DISP,OUTP
...
...
```

Translation of Program HELP

```
.R PAL8
*HELP,LIST<PARAM,HELP
```

Refer to LIST.LS for the save command when loading it with ABSLDR.

3.2 Examples

Before the VC8-E handler can be used, make sure, that program HELP is in core issuing the command R HELP. Now you can use the scope as other non file structured output units like teletype and lineprinter. For instance:

```
.R PAL8
*MAIN,OUTP:<MAIN/C
```

lists the output of CREF after the translation of program MAIN by the PAL8 Assembler.

3.3 Modification to the Editor

A modification which refers to 2.2.1 (entry CHONE) can be done at the program EDIT to the fact, that all output for the teletype and/or the lineprinter is listed on screen. This brings about an enormous increase in speed when editing files. For details refer to the parameter file.

4. Output on Screen by Fortran Programs

4.1 Fortran 2 Programs

4.1.1 Simulated Lineprinter

Using device code 4 in Fortran 2 (free assaignable channel) a program can write on screen via WRITE and FORMAT statements.

Example:

```
      CALL OOPEN('OUTP',0)
      ...
      WRITE (4,1000) IX,IR,F
      ...
1000  FORMAT( ... )
```

As Fortran 2 uses an output buffer a sufficient number of empty lines must be output, if the user has to react on displayed data. Otherwise a partly filled buffer may rest. It is recommended not to use subroutine OCLOSE, as a subsequent call to subroutine OOPEN loads the handler again from the system device into core.

4.1.2 Writing on Arbitrary Positions

Remember, that the handler, if entry DISP is used, starts the display at the position, which is given by the variables XPOS and YPOS of program HELP resident in core. Thus the display starts either where the previous program has stopped or at the position, which is given by an alteration of XPOS and YPOS in core.

Subroutine LOXY

A call to subroutine LOXY (Fortran 2) with two integer arguments for the choosen x and y position modifies variables XPOS and YPOS in corev respectively. Now you can inscribe a graphical display in either of two ways:

(a) a Fortran 2 program uses subroutine LOXY to set the beam at the choosen position and the text is written afterwards by a WRITE statement. Terminate with CALL OCLOSE, as outputting empty lines may force to execute FF, i.e. to erase screen.

(b) a Fortran 2 program reads the x and y position from the keyboard to call subroutine LOXY. The PIP is used to read the text from the keyboard:

```
.R PIP
*DISP: TTY:
```

Modifications, which are to be done to subroutine LOXY, depending on the locations, into which program HELP has been loaded, can be found in the parameter file.

4.2 Output by Fortran 4 Programs

Running Fortran 4 programs, which output text or data on screen, use the Fortran run time system FRTS. It occupies locations in the highest available field to store buffers and handlers there. It depends on the Fortran program, which locations are free; it is never possible to load program HELP into locations 7000-7777.

As there is no subroutine in Fortran 4, which can be compared with subroutine OCLOSE of Fortran 2, a sufficient number of empty lines must be written out to output even partly filled buffers.

Appendix A

```
        /PARAMETER FILE
        /FOR THE VC8E HANDLER AND PROGRAM HELP
        /
        /
        /
        /
        /MAX-PLANCK-INSTITUT FUER PSYCHIATRIE
        /STEPHAN V. BECHTOLSHEIM, ULRICH GSCHREI
        /KRAEPELINSTR. 2
        /D-8000 MUENCHEN 40
        /WESTERN GERMANY
        /
        /
        /VERSION A
        /
        /DATE: SATURDAY APRIL  5, 1975
        /
        /VARIABLES      TRANSLATE
        /               HEL
        /               STRTHL
        /               SIZE
        /               ERFUNC
        /               FFAUTO
        /               BUFFFF
        /MUST BE EDITED BY THE USER
        /
0000    TRANSLATE=0
        /TRANSLATE=0    THE HANDLER IS TRANSLATED
        /TRANSLATE=1    PROGRAM HELP IS TRANSLATED
        /
        /
        /
        /
0030    HEL=30          /FIELD OF PROGRAM HELP. NORMALLY
                        /THE HIGHEST AVAILABLE FIELD
        /
        /
3000    STRTHL=3000     /STARTING ADDRESS OF PROGRAM HELP.
                        /MUST BE AT A PAGE BOUNDARY
```



```
        /VC8E HANDLER FOR OS/8
        /
        /
        /MAX-PLANCK-INSTITUT FUER PSYCHIATRIE
        /STEPHAN V. BECHTOLSHEIM, ULRICH GSCHREI
        /KRAEPELINSTR. 2
        /D-8000 MUENCHEN 40
        /WESTERN GERMANY
        /
        /DATE: SATURDAY,  APRIL 5, 1975
        /
        /VERSION A
        /
        /
2000    FIXMRI INC=2000
                        /ISZ USED AS INSTRUCTION TO INCREMENT
                        /THE CONTENT OF A LOCATION
                        /THE SKIP INSTRUCTION IS NOT USED
7300    CLAL=7300       /CLEAR AC AND LINK
7301    AC0001=7301     /AC:=+1
7330    AC4000=7330     /AC:=4000
```

```
                        /HEADER BLOCK
        0000            *0
00000   7776            -2              /2 ENTRIES
                        /
                        /
                        /
                        /
                        / ***** O U T P *****
                        /FIRST ENTRY CALLED OUTP
                        /ERASE AND INITIALIZE SCREEN
00001   2603            DEVICE VC8E     /GROUP NAME
00002   7005
00003   1725            DEVICE OUTP     /DEVICE NAME
00004   2420
00005   1500            1500            /DEVICE CONTROL BLOCK:
                                        /WRITE ONLY
                                        /DEVICE CODE NUMBER 50
00006   0000            OUTP-BEGIN      /ENTRY POINT WORD:
                                        /ONE PAGE HANDLER
                                        /RELATIVE ENTRY
00007   0000            0;0
00010   0000
                        /
                        /
                        /
                        /
                        / ***** D I S P *****
                        /SECOND ENTRY CALLED DISP
                        /SCREEN IS NOT ERASED ON FIRST ENTRY
00011   2603            DEVICE VC8E     /GROUP NAME
00012   7005
00013   0411            DEVICE DISP     /DEVICE NAME
00014   2320
00015   1510            1510            /DEVICE CONTROL BLOCK:
                                        /WRITE ONLY
                                        /DEVICE CODE NUMBER 51
00016   0113            DISP-BEGIN      /ENTRY POINT WORD:
                                        /ONE PAGE HANDLER
                                        /RELATIVE ENTRY
00017   0000            0;0
00020   0000
```



```
                        /BODY OF THE HANDLER
        0200                *200
                        BEGIN,
                        OUTP,
00200   0000            ENTRY,  0               /ENTRYPOINT TO HANDLER OUTP
00201   7300                    CLAL
00202   2324                    ISZ FIRST       /SKIP ON FIRST ENTRY
00203   7301                    AC0001          /HANDLER HAS BEEN CALLED
                                                /ONCE ALREADY
00204   3326                    DCA MQHELP      /MQ=0: IF HELP IS CALLED
                                                /FIRST ERASE SCREEN
00205   7330            START,  AC4000
00206   0600                    AND I ENTRY     /READ OR WRITE
00207   7004                    RAL             /LINK =0 YOU TRIED TO READ
00210   1600                    TAD I ENTRY
00211   0377                    AND (3700       /WORD COUNT=NUMBER OF
                                                /128-WORD-RECORDS
00212   7040                    CMA
00213   3322                    DCA WC          /-(WC+1) STORED AS WORDCOUNT
00214   6214                    RDF             /READ DATA FIELD =
                                                /INSTRUCTION FIELD
                                                /OF CALLING PROGRAM
00215   1376                    TAD (CDF CIF
00216   3305                    DCA EXCIDF      /INSTRUCTION CIF CDF
                                                /ON RETURN
00217   1600                    TAD I ENTRY     /FIRST ARGUMENT
                                                /LOADED A THIRD TIME
00220   0375                    AND (0070       /MEMORY FIELD BUFFER
00221   1374                    TAD (CDF
00222   3240                    DCA BUFCDF      /CDF FOR BUFFER
00223   2200                    INC ENTRY
00224   1600                    TAD I ENTRY     /GET SECOND ARGUMENT
00225   3323                    DCA CA          /BUFFER ADDRESS
00226   2200                    INC ENTRY       /STARTING BLOCKNUMBER
                                                /IS IGNORED
00227   2200                    INC ENTRY       /ENTRY HOLDS
                                                /ERROR RETURN ADDRESS
00230   7420                    SNL
00231   5307                    JMP ERROR       /YOU TRIED TO READ
00232   1326                    TAD MQHELP
00233   7521                    SWP             /MQ LOADED FOR CALL
                                                /OF PROGRAM HELP
00234   7300                    CLAL
00235   2322            READ,   ISZ WC          /UPDATE WORDCOUNT
00236   5248                    JMP BUFCDF
00237   5303                    JMP RETURN      /BUFFER EXHAUSTED
```

```
                /UNPACK CHARACTERS
00240  7402  BUFCDF, HLT             /CDF BUFFER
00241  7300         CLAL
00242  1723         TAD I CA         /GET FIRST WORD IN BUFFER
00243  0373         AND (0377
00244  3274         DCA CH1          /STORE FIRST CHARACTER
00245  1723         TAD I CA
00246  0372         AND (7400
00247  3276         DCA CH3          /STORE LEFT HALF
                                     /OF THIRD CHARACTER
00250  2323         INC CA
00251  1723         TAD I CA
00252  0373         AND (0377
00253  3275         DCA CH2          /STORE SECOND CHARACTER
00254  1723         TAD I CA
00255  0372         AND (7400        /RIGHT HALF
                                     /OF THIRD CHARACTER
00256  7012         RTR
00257  7012         RTR
00260  1276         TAD CH3
00261  7012         RTR
00262  7012         RTR
00263  3276         DCA CH3          /STORE THIRD CHARACTER
00264  2323         INC CA
00265  6036         KRB
00266  1325         TAD CTRLC
00267  7650         SNA CLA          /SKIP IF NOT CTRL/C TYPED
00270  5311         JMP MONIT        /CALL MONITOR
00271  6201         CDF 00
00272  6232         CIF HEL
00273  4771         JMS I (STRTHL+4  /CALL PROGRAM HELP
00274  0000  CH1,    0
00275  0000  CH2,    0
00276  0000  CH3,    0
00277  5303  RETA,   JMP RETURN       /RETURN A FROM HELP
                                     /CTRL/Z FOUND IN BUFFER
                                     /OR READ FROM KEYBOARD
00300  7301  RETB,   AC0001          /RETURN B FROM HELP
                                     /TO AVOID FURTHER
                                     /FORMFEEDS
00301  7521         SWP
00302  5235         JMP READ         /UNPACK NEXT CHARACTERS
```



```
                /RETURN FROM HANDLER
00303  7300  RETURN, CLAL
00304  2200         INC ENTRY        /NO ERROR RETURN
00305  7402  EXCIDF, HLT             /CIF CDF CALLING PROGRAM
00306  5600         JMP I ENTRY      /RETURN
00307  7330  ERROR,  AC4000          /ERROR RETURN AC=4000
00310  5305         JMP EXCIDF
00311  6203  MONIT,  CIF CDF 00      /CTRL/C CALLS MONITOR
00312  5770         JMP I (7600
               /
               /
               /
               /
00313  0000  DISP,   0                /ENTRYPOINT TO HANDLER DISP
00314  7300         CLAL
00315  1313         TAD DISP
00316  3200         DCA ENTRY        /ENTRY HOLDS ADDRESS
                                     /OF FIRST ARGUMENT
00317  7301         AC0001
00320  3326         DCA MQHELP       /ERASE FUNCTION IS
                                     /SUPRESSED ON FIRST
                                     /ENTRY TO HELP
00321  5205         JMP START        /JUMP TO GET ARGUMENTS
               /
               /
               /
               /
00322  0000  WC,     0                /WORD COUNT
                                     /COUNTS 2 WORDS
                                     /=3 CHARACTERS
00323  0000  CA,     0                /PRESENT ADDRESS IN BUFFER
00324  7777  FIRST,  -1               /TO CONTROL FIRST ENTRY
                                     /TO THE HANDLER
00325  7575  CTRLC,  -203            /NEGATIVE VALUE OF CTRL/C
00326  0000  MQHELP, 0                /VALUE OF MQ, IF HELP
                                     /IS CALLED FIRST
                                     /=0 FORCES ERASE
00370  7600
00371  3004
00372  7400
00373  0377
00374  6201
00375  0070
00376  6203
00377  3700
               $
```

ERRORS DETECTED: 0
LINKS GENERATED: 0

```
AC0001 7301
AC4000 7330
BEGIN  0200
BUFCDF 0240
CA     0323
CH1    0274
CH2    0275
CH3    0276
CLAL   7300
CTRLC  0325
DISP   0313
ENTRY  0200
ERROR  0307
EXCIDF 0305
FIRST  0324
HEL    0030
INC    2000
MONIT  0311
MQHELP 0326
OUTP   0200
READ   0235
RETA   0277
RETB   0300
RETURN 0303
START  0205
STRTHL 3000
TRANSL 0000
WC     0322
```

Apendix B

```
        /PARAMETER FILE
        /FOR THE VC8E HANDLER AND PROGRAM HELP
        /
        /
        /
        /
        /
        /MAX-PLANCK-INSTITUT FUER PSYCHIATRIE
        /STEPHAN V. BECHTOLSHEIM, ULRICH GSCHREI
        /KRAEPELINSTR. 2
        /D-8000 MUENCHEN 40
        /WESTERN GERMANY
        /
        /
        /VERSION A
        /
        /DATE: SATURDAY APRIL  5, 1975
        /
        /VARIABLES      TRANSLATE
        /               HEL
        /               STRTHL
        /               SIZE
        /               ERFUNC
        /               FFAUTO
        /               BUFFFF
        /MUST BE EDITED BY THE USER
        /
0001    TRANSLATE=1
        /TRANSLATE=0    THE HANDLER IS TRANSLATED
        /TRANSLATE=1    PROGRAM HELP IS TRANSLATED
        /
        /
        /
        /
0030    HEL=30          /FIELD OF PROGRAM HELP. NORMALLY
                        /THE HIGHEST AVAILABLE FIELD
        /
        /
3000    STRTHL=3000     /STARTING ADDRESS OF PROGRAM HELP.
                        /MUST BE AT A PAGE BOUNDARY
```



```
        /
        /
        /
        /
0001    ERFUNC=1
        /ERFUNC=0   ERASE FUNCTION NOT INSTALLED
        /       =1  ERASE FUNCTION INSTALLED
        /
        /
0000    FFAUTO=0
        /FFAUTO=0   FORMFEED EXECUTED ONLY,
        /           IF SPACE TYPED
        /       =1  FORMFEED EXECUTED AUTOMATICALLY
        /
        /
0000    BUFFFF=0
        /BUFFFF=0   FORMFEEDS READ IN BUFFER
        /           ARE REPLACED BY LINEFEED
        /       =1  FORMFEEDS READ IN BUFFER
        /           ARE EXECUTED
        /
        /
0002    SIZE=2
        /CHARACTER SIZE, 5X7 DOT MATRIX
        /1 SMALLEST SIZE, 3 MAXIMAL SIZE
        /
3777    ENDHL=STRTHL+777
6231    CDFHEL=CDF HEL
6232    CIFHEL=CIF HEL
3700    CHADR=CHONE
3356    ADXPOS=XPOS
3357    ADYPOS=YPOS
```

```
        /SAVE COMMAND FOR PROGRAM HELP:
        /SAVE SYS:HELP STRTHL:ENDHL;STRTHL=JOB STATUS WORD
        /THE NUMBER OF THE FIELD MUST BE INCLUDED
        /
        /
        /
        /SUBROUTINE LOXY
        /WHEN TRANSLATING SUBROUTINE LOXY
        /MODIFY ADXPOS, ADYPOS, CDFHEL RESPECTIVELY
        /
        /
        /
        /A SEQUENCE OF INSTRUCTIONS TO PRINT ON THE TELETYPE
        /LIKE:  TAD CHAR        /GET CHARACTER
        /       TLS             /AND PRINT IT
        /       TSF             /WAIT ON FLAG
        /       JMP .-1
        /CAN BE REPLACED BY:
        /       TAD CHAR
        /       CIFHEL
        /       JMS I (CHOAR
        /SO ALL OUTPUT WILL DIRECTED TO THE SCOPE
        /(THIS MODIFICATION HAS BEEN DONE TO THE EDITOR
        /AND IS DESCRIPED BELOW)
        /
        /
        /
        /
        /MODIFICATION OF THE EDITOR
        /
        /(A) ALL PRINTING EXCEPT FOR THE V-COMMAND IS
        /DIRECTED TO SCOPE
        /
        /       GET SYS:EDIT
        /       ODT
        /       1222/6046 CIFHEL
        /       1223/6041 4625
        /       1224/5223 5621
        /       1225/7200 CHADR
        /       CTRL/C
        /       SAVE SYS:EDIT1
        /
        /(B) THE V-COMMAND LISTS THE BUFFER ON SCREEN
        /
        /       GET SYS:EDIT
        /       ODT
        /       2713/6666 CIFHEL
        /       2714/6661 4716
        /       2715/5314 5712
        /       2716/7200 CHADR
        /       CTRL/C
        /       SAVE SYS:EDIT2
        /
        /
        /
```



```
        /HELP
        /ASSISTANT PROGRAM FOR VC8E HANDLER
        /
        /
        /
        /MAX-PLANCK-INSTITUT FUER PSYCHIATRIE
        /STEPHAN V. BECHTOLSHEIM, ULRICH GSCHREI
        /KRAEPELINSTR. 2
        /D-8000 MUENCHEN 40
        /WESTERN GERMANY
        /
        /
        /VERSION A
        /
        /DATE: SATURDAY  APRIL 5, 1975
        /
        /
        /
2000    FIXMRI INC=2000
                        /ISZ USED AS INSTRUCTION TO INCREMENT
                        /THE CONTENT OF A LOCATION
                        /THE SKIP INSTRUCTION IS NOT USED
5400    FIXMRI EXIT=5400    /RETURN FROM A SUBROUTINE
6052    DISD=6052       /SKIP ON DONE FLAG
6053    DILX=6053       /LOAD X-REGISTER
6054    DILY=6054       /LOAD Y-REGISTER
6055    DIXY=6055       /INTENSIFY
6056    DILE=6056       /LOAD ENABLE REGISTER
6057    DIRE=6057       /READ ENABLE REGISTER
7240    AC7777=7240     /AC:=-1
7300    CLAL=7300       /CLEAR AC AND LINK
7621    CAM=7621        /CLEAR AC AND MQ
```

```
                FIELD HEL%10
0003            *STRTHL
3000    /
        /
        /START ADDRESS LOAD IS WITH IMMEDIATE RETURN
        /TO MONITOR AFTER EXECUTION, I.E. THE RUN
        /COMMAND ACTS AS A LOAD COMMAND
        /OF PROGRAM HELP
        /
        /
        /
        HELP,
33000 7300      LOAD,   CLAL
33001 6203      MONIT,  CDF CIF 00
33002 5603              JMP I D7600     /CALLS MONITOR
33003 7600      D7600,  7600
33004 0000      HAND,   0               /ENTRY FOR HANDLER
33005 7300              CLAL
33006 1604              TAD I HAND      /GET FIRST CHARACTER
33007 3224              DCA CH1         /AND STORE IT
33010 2204              INC HAND
33011 1604              TAD I HAND      /GET SECOND CHARACTER
33012 3226              DCA CH2         /AND STORE IT
33013 2204              INC HAND
33014 1604              TAD I HAND      /GET THIRD CHARACTER
33015 3230              DCA CH3         /AND STORE IT
33016 2204              INC HAND        /HAND HOLDS ADDRESS
                                        /OF RETURN A
                                        /IN HANDLER
33017 6231              CDF HEL         /ALL ARGUMENTS FROM HANDLER
                                        /ARE FETCHED NOW
33020 7521              SWP
33021 7650              SNA CLA         /MQ=0 FORCES ERASE
33022 4253              JMS ERASE
        /
        /
        /
33023 4777'             JMS CHAR
33024 0000      CH1,    0
33025 4777'             JMS CHAR
33026 0000      CH2,    0
33027 4777'             JMS CHAR
33030 0000      CH3,    0
33031 2204              INC HAND        /HAND HOLDS ADDRESS OF
                                        /RETURN B TO HANDLER
33032 6203      CTZFND, CDF CIF 00      /JUMP TO THIS LOCATION
                                        /OCCURS, IF CTRL/Z
                                        /READ FROM KEYBOARD
                                        /OR IN BUFFER
33033 5604              EXIT HAND       /RETURN TO HANDLER
```



```
        /SUBROUTINE TO EXECUTE LINEFEED
33034 0000      LFCR,   0
33035 1332              TAD XINI
33036 6053              DILX
33037 3776'             DCA XPOS        /XPOS LOADED
                                        /WITH INITIAL VALUE
33040 1775'             TAD YPOS
33041 1342              TAD NEWLNE      /ADD INCREMENT FOR NEWLINE
33042 6054              DILY
33043 3775'             DCA YPOS        /YPOS UPDATED
33044 7240              AC7777
33045 3331              DCA POSNUM      /POSNUM=-1
33046 1775'             TAD YPOS
33047 1335              TAD YMIN
33050 7710              SPA CLA         /TOO MANY LINES?
33051 4272              JMS FF          /YES, CALL FORMFEED
33052 5634              EXIT LFCR
        /
        /
        /
        /SUBROUTINE TO ERASE SCREEN
        /AND TO RESET BEAM
33053 0000      ERASE,  0
                        IFNZRO ERFUNC <
33054 1374              TAD (30         /ERASE AND STORE MODE
33055 6056              DILE
33056 6057      WAIT,   DIRE            /WAIT FOR DONE FLAG
33057 7700              SMA CLA         /SKIP ON DONE FLAG
33060 5256              JMP WAIT
                        >
33061 7240              AC7777
33062 3331              DCA POSNUM      /POSNUM=-1
33063 1332              TAD XINI
33064 6053              DILX
33065 3776'             DCA XPOS        /XPOS LOADED
                                        /WITH INITIAL VALUE
33066 1334              TAD YINI
33067 6054              DILY
33070 3775'             DCA YPOS        /SAME TO YPOS
33071 5653              EXIT ERASE      /RETURN FROM ERASE
```

```
        IFNZRO FFAUTO <
/SUBROUTINE TO EXECUTE AN AUTOMATIC AND
/DELAYED FORMFEED
FF,     0
        CLAL            /INITIALIZE COUNTERS
        TAD SECOND
        DCA DELSEC      /DELAY IN SECONDS
LOOP,   TAD INDEL1
        DCA DEL1        /DELAY COUNTER 1 INITIALIZED
LOOPIN, ISZ DEL2        /BUSY WAITING
        JMP .-1
        KRB
        TAD CTRLC
        SNA CLA
        JMP MONIT       /CALL MONITOR
        KRB
        TAD CTRLZ
        SZA CLA
        JMP WZ
        JMS ERASE
        JMP CTZFND      /CTRL/Z TYPED: USE RETURN A
                        /IN HANDLER
WZ,     KRB
        TAD CTRLS
        SNA CLA
        JMP STOP        /CTRL/S TYPED:
                        /DELAY UNTIL SPACE TYPED
        ISZ DEL1
        JMP LOOPIN
        ISZ DELSEC
        JMP LOOP        /DELAY FOR A FURTHER SECOND
        JMP FFEND       /ERASE TIME EXCEEDED
STOP,   KRB
        TAD SPACE       /IF SPACE IS TYPED,
                        /ERASE IS EXECUTED
        SZA CLA
        JMP STOP
FFEND,  JMS ERASE       /RETURN FROM FORMFEED
        EXIT FF
DELSEC, 0               /COUNTS DELAY IN SECONDS
DEL1,   0               /DELAY COUNTER
DEL2,   0               /DELAY COUNTER
SECOND, -2              /NEGATIVE VALUE OF DELAY
                        /IN SECONDS
INDEL1, -76             /INITIAL VALUE FOR DEL1
CTRLS,  -223            /NEGATIVE VALUE OF CTRL/S
        >
```



```
        IFZERO FFAUTO <
/SUBROUTINE TO EXECUTE FORMFEED
/IF SPACE TYPED ON KEYBOARD
33072  0000  FF,     0
33073  6030  AGAIN,  KCF
33074  6031         KSF
33075  5274         JMP .-1
33076  6036         KRB
33077  1340         TAD CTRLC
33100  7650         SNA CLA
33101  5201         JMP MONIT       /CTRL/C TYPED, CALL MONITOR
33102  6036         KRB
33103  1337         TAD SPACE
33104  7640         SZA CLA
33105  5310         JMP M1
33106  4253         JMS ERASE       /SPACE TYPED
                                    /EXECUTE FORMFEED AND ERASE
33107  5672         EXIT FF
33110  6036  M1,    KRB
33111  1341         TAD CTRLZ
33112  7640         SZA CLA
33113  5273         JMP AGAIN       /NO RELEVANT CHARACTER
                                    /READ FROM KEYBOARD
33114  4253         JMS ERASE       /CTRL/Z TYPED
                                    /RETURN AS IF
                                    /CTRL/Z  FOUND IN BUFFER
                                    /ERASE SCREEN
33115  5232         JMP CTZFND      /RETURN A TO HANDLER
                    >
```

```
                    /SUBROUTINE TO DISPLAY A SPACE:
                    /ONLY THE X-POSITION IS INCREMENTED
                    /
33116  0000  SPADIS, 0
33117  1776'        TAD XPOS
33120  1333         TAD XINCR
33121  3776'        DCA XPOS          /XPOS INCREMENTED
33122  5716         EXIT SPADIS
                    /
                    /
                    /
                    /SUBROUTINE WHICH CALLS SUBROUTINE LFCR
                    /IF LINE TOO LONG
33123  0000  TOOLON, 0
33124  1776'        TAD XPOS
33125  1336         TAD XEND
33126  7700         SMA CLA
33127  4234         JMS LFCR          /LINE WAS TOO LONG
33130  5723         EXIT TOOLON
                    /
                    /
                    /
                    /
33131  0000  POSNUM, 0                /NUMBER OF POSITION OF
                                      /NEXT CHARACTER, WHICH
                                      /IS DISPLAYED
33132  7001  XINI,  7001              /INITIAL VALUE OF X
33133  0015  XINCR, SIZE^6+1
33134  0730  YINI,  0730              /INITIAL VALUE OF Y
33135  0777  YMIN,  0777              /TO CONTROL THE MAXIMUM
                                      /NUMBER OF LINES
                                      /ON SCREEN
33136  7013  XEND,  -765              /TO CONTROL THE LENGTH
                                      /OF A LINE
33137  7540  SPACE, -240              /8 BIT ASCII SPACE
33140  7575  CTRLC, -203              /NEGATIVE VALUE OF CTRL/C
33141  7546  CTRLZ, -232              /NEGATIVE VALUE OF CTRL/Z
33142  7750  NEWLNE, -SIZE^10-10      /DETERMINES DISTANCE
                                      /OF LINES
```



```
33174  0030
33175  3357
33176  3356
33177  3200
       3200         PAGE
                    /CHAR DETECTS CONTROL CHARACTERS
                    /ORDINARY CHARACTERS ARE DISPLAYED
                    /FIRST LOOK FOR CONTROL CHARACTERS
33200  0000  CHAR,  0
33201  2777'        INC POSNUM
33202  7000         NOP
33203  1600         TAD I CHAR        /GET CHARACTER
33204  1776'        TAD SPACE
33205  7640         SZA CLA
33206  5211         JMP W1
33207  4775'  SPSP, JMS SPADIS        /SPACE IS FOUND
33210  5341         JMP CHREND
33211  1600  W1,    TAD I CHAR
33212  1351         TAD FORMFD
33213  7640         SZA CLA           /SKIP ON FORMFEED
33214  5217         JMP W2
                    IFZERO BUFFFF <
33215  4774'        JMS LFCR          /FF REPLACED BY LF
                    >
                    IFNZRO BUFFFF <
                    JMS FF            /FF IS EXECUTED
                    >
33216  5342         JMP BACK          /RETURN FROM CHAR
33217  1600  W2,    TAD I CHAR
33220  1773'        TAD CTRLZ
33221  7650         SNA CLA           /SKIP IF NO CTRL/Z FOUND
33222  5772'        JMP CTZFND        /CTRL/Z FOUND
                                      /USE RETURN A IN HANDLER
33223  1600         TAD I CHAR
33224  1350         TAD LNEFD
33225  7650         SNA CLA           /SKIP ON NO LINEFEED
33226  5342         JMP BACK          /IGNORE LINEFEED
33227  1600         TAD I CHAR
33230  1352         TAD CARRET
33231  7640         SZA CLA           /SKIP, IF CR FOUND
33232  5235         JMP W3
33233  4774'        JMS LFCR          /EXECUTE LINEFEED
                                      /AND CARRIAGE RETURN
33234  5342         JMP BACK
33235  1600  W3,    TAD I CHAR
33236  1371         TAD (-12          /OCTAL 012 FROM PIP
                                      /IS IGNORED. OCCURS,
                                      /IF DIRECTORIES ARE LISTED
33237  7650         SNA CLA
33240  5342         JMP BACK
33241  1600         TAD I CHAR
33242  1346         TAD LETR
33243  7650         SNA CLA           /IGNORE LEADER/TRAILER
33244  5342         JMP BACK
33245  1600         TAD I CHAR
```

```
33246  1347          TAD TABU
33247  7640          SZA CLA          /IS IT TABULATOR?
33250  5260          JMP W4           /NO
33251  1777'  MOSA,  TAD POSNUM       /SPACES UNTIL
                                      /POSNUM=0 MOD 8
33252  0370          AND (0007
33253  7650          SNA CLA
33254  5207          JMP SPSP         /ONE MORE
33255  4775'         JMS SPADIS       /DISPLAY A SPACE
33256  2777'         INC POSNUM
33257  5251          JMP MOSA
33260  1600  W4,     TAD I CHAR
33261  1353          TAD RUBOUT
33262  7650          SNA CLA          /IS IT RUBOUT
33263  5341          JMP CHREND       /YES: IGNORE IT
               /IT WAS NOT A CONTROL-CHARACTER
               /
               /CALCULATE ADDRESS OF THE FIRST DISPLAY WORD
               /OF THE CHARACTER
33264  1600  DISCH,  TAD I CHAR
33265  0367          AND (0077        /6 BIT ASCII
33266  3344          DCA CHY
33267  1344          TAD CHY
33270  1344          TAD CHY
33271  1344          TAD CHY          /AC:=CHY*3
33272  1366          TAD (TABLE       /ADD BASIC ADDRESS OF TABLE
33273  3344          DCA CHY          /ADDRESS OF FIRST
                                      /DISPLAY WORD
```



```
               /DISPLAY CHARACTER
33274  1365          TAD (-4
33275  3345          DCA CTW          /INITIAL VALUE OF CTW IS -4
33276  7240          AC7777
33277  3354          DCA UNCT         /UNCT:=-1
33300  7621          CAM
33301  1356  NEWX,   TAD XPOS
33302  1364          TAD (SIZE        /XPOS IS INCREMENTED
33303  6053          DILX
33304  3356          DCA XPOS
33305  1363          TAD (-7
33306  3355          DCA YCNT         /INITIALIZE COUNTER FOR Y
33307  1357          TAD YPOS
33310  7521  CYCL,   SWP              /YPOS TO MQ
33311  2354          ISZ UNCT         /SKIP IF NO BIT
                                      /IN THE DISPLAY WORD
                                      /IS LEFT
33312  5324          JMP CTUE         /THERE ARE BITS LEFT
33313  2345          ISZ CTW          /SKIP, IF NO WORD LEFT
33314  5320          JMP DXM          /JUMP TO GET NEXT WORD
                                      /OF TABLE
33315  2356          INC XPOS         /TO HAVE SPACE
                                      /BETWEEN TWO CHARACTERS
33316  7000          NOP
33317  5341          JMP CHREND       /JUMP TO THE END OF CHAR
33320  1362  DXM,    TAD (-14
33321  3354          DCA UNCT         /INITIALIZE UNCT
33322  1744          TAD I CHY        /LOAD AC
                                      /WITH NEW DISPLAY-WORD
33323  2344          INC CHY          /CHY POINTS TO THE NEXT
                                      /DISPLAY WORD
33324  7104  CTUE,   CLL RAL          /LINK LOADED BY NEXT BIT
                                      /OF THE DISPLAY WORD
33325  7521          SWP
33326  7420          SNL              /LINK=L SKIP TO INTENSIFY
33327  5334          JMP DNIX         /DO NOT INTENSIFY
33330  6054          DILY             /AC HOLDS VALUE OF YPOS
33331  6052          DISD
33332  5331          JMP .-1
33333  6055          DIXY
33334  1364  DNIX,   TAD (SIZE       /YPOS IN AC IS INCREMENTED
33335  2355          ISZ YCNT         /SKIP IF 7 POINTS
                                      /IN THE DIRECTION
                                      /OF THE Y-AXIS
                                      /HAVE BEEN DISPLAYED
33336  5310          JMP CYCL
33337  7300          CLAL
33340  5301          JMP NEWX         /GET NEW VALUE FOR XPOS
33341  4761'  CHREND, JMS TOOLON       /CHARACTER DISPLAYED
                                      /LINE TOO LONG?
33342  2200  BACK,   INC CHAR         /RETURN ADDRESS OF CHAR
33343  5600          EXIT CHAR
```

```
33344  0000  CHV,    0              /ADDRESS OF DISPLAY WORD
33345  0000  CTW,    0              /COUNTS NUMBER
                                    /OF DISPLAY WORDS
33346  7600  LETR,   -200           /NEGATIVE VALUE OF
                                    /LEADER/TRAILER CODE
33347  7567  TABU,   -211           /NEGATIVE VALUE OF TABULATOR
33350  7566  LNEFD,  -212           /NEGATIVE VALUE OF LINEFEED
33351  7564  FORMFD, -214           /NEGATIVE VALUE OF FORMFEED
33352  7563  CARRET, -215           /NEGATIVE VALUE OF
                                    /CARRIAGE RETURN
33353  7401  RUBOUT, -377           /NEGATIVE VALUE OF RUBOUT
33354  0000  UNCT,   0              /COUNTS BITS OF DISPLAY WORD
33355  0000  YCNT,   0              /COUNTS THE POINTS
                                    /IN THE DIRECTION
                                    /OF THE Y-AXIS
33356  0000  XPOS,   0              /X-POSITION OF BEAM
33357  0000  YPOS,   0              /Y-POSITION OF BEAM
```



```
33361  3123
33362  7764
33363  7771
33364  0002
33365  7774
33366  3400
33367  0077
33370  0007
33371  7766
33372  3032
33373  3141
33374  3034
33375  3116
33376  3137
33377  3131
       3400         PAGE
                    /TABLE CONTAINING THE DISPLAY WORDS FOR EACH
                    /CHARACTER.  A  5 X 7  POINT MATRIX IS USED.
                    /SO THREE DISPLAY WORDS BELONG TO EACH CHARACTER
                    /REFERENCE: MAX-PLANCK-INSTITUT FUER PHYSIK
                    /UND ASTROPHYSIK, A.J. FLAYELL
                    /D-8000 MEUNCHEN, WESTERN GERMANY
                    TABLE,
33400  0000  0000;   0000;   0000;   /00 SPACE
33401  0000
33402  0000
33403  7604  7604;   4211;   1370    /01 A
33404  4211
33405  1370
33406  4077  4077;   7114;   4554    /02 B
33407  7114
33410  4554
33411  3720  3720;   3014;   0504    /03 C
33412  3014
33413  0504
33414  4077  4077;   7014;   0574    /04 D
33415  7014
33416  0574
33417  7762  7762;   3114;   4602    /05 E
33420  3114
33421  4602
33422  7742  7742;   2110;   4402    /06 F
33423  2110
33424  4402
33425  3720  3720;   3015;   0744    /07 G
33426  3015
33427  0744
33430  7742  7742;   0100;   4376    /10 H
33431  0100
33432  4376
```

```
33433  0020  0020;  3774;  0400  /11 I
33434  3774
33435  0400
33436  2020  2020;  3014;  0576  /12 J
33437  3014
33440  0576
33441  7742  7742;  0242;  1202  /13 K
33442  0242
33443  1202
33444  7760  7760;  1004;  0200  /14 L
33445  1004
33446  0200
33447  7740  7740;  4140;  1376  /15 M
33450  4140
33451  1376
33452  7741  7741;  4103;  0376  /16 N
33453  4103
33454  0376
33455  3720  3720;  3014;  0574  /17 O
33456  3014
33457  0574
33460  7742  7742;  2110;  4414  /20 P
33461  2110
33462  4414
33463  3720  3720;  3212;  0674  /21 Q
33464  3212
33465  0674
33466  7742  7742;  2312;  4614  /22 R
33467  2312
33470  4614
33471  2322  2322;  3114;  4544  /23 S
33472  3114
33473  4544
33474  0040  0040;  3770;  0402  /24 T
33475  3770
33476  0402
33477  3760  3760;  1004;  0176  /25 U
33500  1004
33501  0176
33502  0356  0356;  1003;  4016  /26 V
33503  1003
33504  4016
33505  7750  7750;  0302;  0376  /27 W
33506  0302
33507  0376
33510  6145  6145;  0101;  2306  /30 X
33511  0101
33512  2306
```



```
33513  0141  0141;  1700;  2006  /31 Y
33514  1700
33515  2006
33516  6064  6064;  3114;  2606  /32 Z
33517  3114
33520  2606
33521  0037  0037;  7014;  0400  /33 [
33522  7014
33523  0400
33524  0101  0101;  0101;  0100  /34 BACKSLASH
33525  0101
33526  0100
33527  0020  0020;  3017;  7400  /35 ]
33530  3017
33531  7400
33532  0200  0200;  5770;  1010  /36 ^
33533  5770
33534  1010
33535  0407  0407;  0520;  4020  /37 BACK ARROW
33536  0520
33537  4020
33540  0000  0000;  0000;  0000  /40 SPACE
33541  0000
33542  0000
33543  0000  0000;  1370;  0000  /41 !
33544  1370
33545  0000
33546  0000  0000;  6000;  1400  /42 "
33547  6000
33550  1400
33551  1237  1237;  6247;  7450  /43 #
33552  6247
33553  7450
33554  2212  2212;  5772;  5044  /44 $
33555  5772
33556  5044
33557  2144  2144;  610C;  2304  /45 %
33560  6106
33561  2304
33562  3322  3322;  3262;  0240  /46 &
33563  3262
33564  0240
33565  0000  0000;  0030;  0000  /47 '
33566  0030
33567  0000
33570  0007  0007;  0424;  0400  /50 (
33571  0424
33572  0400
```

```
33573  0020  0020;  2421;  6000    /51 )
33574  2421
33575  6000
33576  4452  4452;  4342;  5222    /52 *
33577  4342
33600  5222
33601  0402  0402;  0760;  4020    /53 +
33602  0760
33603  4020
33604  0024  0024;  0600;  0000    /54 ,
33605  0600
33606  0000
33607  0402  0402;  0100;  4020    /55 -
33610  0100
33611  4020
33612  0030  0030;  1400;  0000    /56 .
33613  1400
33614  0000
33615  2004  2004;  0100;  2004    /57 /
33616  0100
33617  2004
33620  1620  1620;  5024;  1160    /60 0
33621  5024
33622  1160
33623  0020  0020;  5774;  0000    /61 1
33624  5774
33625  0000
33626  4130  4130;  3215;  0634    /62 2
33627  3215
33630  0634
33631  2120  2120;  3114;  4554    /63 3
33632  3114
33633  4554
33634  1405  1405;  0227;  7440    /64 4
33635  0227
33636  7440
33637  2361  2361;  3054;  2562    /65 5
33640  3054
33641  2562
33642  3722  3722;  3114;  4540    /66 6
33643  3114
33644  4540
33645  4050  4050;  2210;  4416    /67 7
33646  2210
33647  4416
33650  3123  3123;  3114;  6544    /70 8
33651  3114
33652  6544
```



```
33653  0322  0322;  3114;  4574    /71 9
33654  3114
33655  4574
33656  0030  0030;  7430;  0000    /72 :
33657  7430
33660  0000
33661  0024  0024;  6630;  0000    /73 ;
33662  6630
33663  0000
33664  0405  0405;  0424;  0400    /74 <
33665  0424
33666  0400
33667  1205  1205;  0241;  2050    /75 =
33670  0241
33671  2050
33672  4050  4050;  4240;  4000    /76 >
33673  4240
33674  4000
33675  0100  0100;  3310;  4414    /77 ?
33676  3310
33677  4414
```

```
33700  0000  CHONE,  0               /ENTRY TO DISPLAY
                                     /A SINGLE CHARACTER
33701  3317          DCA CH4         /AC HOLDS CHARACTER
33702  6214          RDF
33703  1377          TAD (CDF CIF
33704  3321          DCA CDIRET
33705  6231          CDF HEL
33706  1317          TAD CH4
33707  1330          TAD CTRLG
33710  7650          SNA CLA         /IS IT BELL?
33711  5323          JMP CTGFND      /YES
33712  1317          TAD CH4
33713  1776'         TAD CTRLC
33714  7650          SNA CLA         /IS IT CTRL/C?
33715  5775'         JMP MONIT       /YES: CALL MONITOR
33716  4774'         JMS CHAR        /DISPLAY CHARACTER
33717  0000  CH4,    0
33720  7300  RETCHO, CLAL
33721  7402  CDIRET, HLT             /RESTORE DF AND IF
33722  5700          EXIT CHONE
33723  1317  CTGFND, TAD CH4
33724  6046          TLS
33725  6041          TSF
33726  5325          JMP .-1
33727  5320          JMP RETCHO
33730  7571  CTRLG,  -207            /NEGATIVE VALUE OF CTRL/G
33774  3200
33775  3001
33776  3140
33777  6203
```

$

AC7777	7240	MOSA	3251
ADXPOS	3356	M1	3110
ADYPOS	3357	NEWLNE	3142
AGAIN	3073	NEWX	3301
BACK	3342	POSNUM	3131
BUFFFF	0000	RETCHO	3720
CAM	7621	RUBOUT	3353
CARRET	3352	SIZE	0002
CDFHEL	6231	SPACE	3137
CDIRET	3721	SPADIS	3116
CHADR	3700	SPSP	3207
CHAR	3200	STRTHL	3000
CHONE	3700	TABLE	3400
CHREND	3341	TABU	3347
CHV	3344	TOOLON	3123
CH1	3024	TRANSL	0001
CH2	3026	UNCT	3354
CH3	3030	WAIT	3056
CH4	3717	W1	3211
CIFHEL	6232	W2	3217
CLAL	7300	W3	3235
CTGFND	3723	W4	3260
CTRLC	3140	XEND	3136
CTRLG	3730	XINCR	3133
CTRLZ	3141	XINI	3132
CTUE	3324	XPOS	3356
CTW	3345	YCNT	3355
CTZFND	3032	YINI	3134
CYCL	3310	YMIN	3135
DILE	6056	YPOS	3357
DILX	6053		
DILY	6054		
DIRE	6057		
DISCH	3264		
DISD	6052		
DIXY	6055		
DNIX	3334		
DXM	3320		
D7600	3003		
ENDHL	3777		
ERASE	3053		
ERFUNC	0001		
EXIT	5400		
FF	3072		
FFAUTO	0000		
FORMFD	3351		
HAND	3004		
HEL	0030		
HELP	3000		
INC	2000		
LETR	3346		
LFCR	3034		
LNEFD	3350		
LOAD	3000		
MONIT	3001		

```
ERRORS DETECTED: 0
LINKS GENERATED: 26
```

Appendix 6

```
      SUBROUTINE LOXY (IX0,IY0)
C
C     SUBROUTINE TO POSITION BEAM
C     FORTRAN2
C
C
C     MAX-PLANCK-INSTITUT FUER PSYCHIATRIE
C     STEPHAN V. BECHTOLHEIM, ULRICH GSCHREI
C     KRAEPELINSTR.5
C     8 MUENCHEN 40
C     WESTERN GERMANY
C
C     DATE: SATURDAY  APRIL 5, 1975
C
C     VERSION A
C
C
C
C     THE USER MUST EDIT THE VARIABLES ADXPOS AND ADYPOS
C     INCLUDING THE INSTRUCTION CDFHEL ACCORDING TO THE
C     PARAMETER FILE
C
C
C
C
C
C     IX0=X-POSITION OF BEAM
C     IY0=Y-POSITION OF BEAM
C
C
C
      IX=IX0
      IY=IY0
C
C
S OPDEF CLAL    7300    /CLEAR AC AND LINK
S OPDEF RDF     6214    /READ DATA FIELD
S OPDEF CDFHEL  6231    /CDF FIELD OF PROGRAM HELP
S                       /REFER TO THE PARAMETER FILE
S OPDEF DILX    6053    /LOAD X-REGISTER OF VC8E
S OPDEF DILY    6064    /LOAD Y-REGISTER OF VC8E
S       CLAL
S       RDF
S       TAD ICDF
S       DCA CDFCF       /INSTRUCTION CDF
S                       /CURRENT FILED
S       CDFHEL          /CDF FOR LOADING XPOS
S                       /AND YPOS IN PROGRAM HELP
S       TAD \IX
S       DILX
S M,    3661            /DCA I ADXPOS
S       TAD \IY
S       DILY
S N,    3662            /DCA I ADYPOS
S CDFCF,0               /RESTORE CURRENT DATA FIELD
S       CLAL
        RETURN
S ADXPOS,3356           /ADDRESS OF XPOS IN PROGRAM
S                       /HELP. REFER TO THE PARAMETER FILE
S ADYPOS,3357           /SAME TO YPOS
S ICDF, 6201            /CDF
        END
```